
非線形シュレディンガー方程式の数値計算

－素人による素人のための

太田 雅人（埼玉大学理学部数学科）

このノートは、RIMS短期共同研究「非線形双曲型および分散型方程式の研究」
での筆者の 2011年 5月 23日の講演をまとめたものです。数値計算歴 7ヶ月の素人
がこのような講演を行った理由を簡単に説明したいと思います。現在、非線形双
曲型・分散型方程式の数学的研究は世界的に盛んに行われていますが、フランス
や米国に比べ、日本では数値計算を行う非線形双曲型・分散型方程式の数学研究
者の数は極めて少ないように思います。日本でも、反応拡散方程式系や流体方程
式の研究グループでは、数学解析と数値計算を有機的に組み合わせた研究が行わ
れていると思います。また、非線形双曲型・分散型方程式に関しても、数学解析と
数値計算を組み合わせた研究はフランスや米国では盛んに行われています。昨年
度１年間滞在したボルドー第１大学でもそのことを実感しました。
以下では、ボルドー第１大学で勉強してきた、relaxation sheme of Besseと split-

ting method を用いた、非線形シュレディンガー方程式の数値計算について解説し
ます。実際の数値計算では、ソフトウェア Scilab を使っています。Scilab は数値
計算のためのフリーのソフトウェアで http://www.scilab.org/ からダウンロード
できます。また、Google などで「Scilab」で検索すれば、日本語の解説や参考書
などを見つけることができます。Scilab の使い方はとても簡単で、数値計算につ
いて全くの素人であった筆者でも、プログラムの書き方などを Mathieu Colin 氏
に教えてもらい、１週間ほどで使えるようになりました。
数値計算が非線形双曲型・分散型方程式の数学研究に直ちに役に立つかどうか
は分りませんが、趣味程度にでもやってみると楽しいと思います。数値計算を全
くやったことがない人にとっては、ちょっとしたことが分らず、先に進めないこと
があります。そのような場合は、よく知っている人が近くにいれば、その人に教え
てもらうのが一番の解決策です。実際、筆者が短期間で使えるようになったのは、
Mathieu Colin 氏のお陰です。もし教えてくれる人が近くにいない場合は、メール
などで筆者にご相談ください。数値計算もできる、非線形双曲型・分散型方程式
の数学研究者が一人でも増えることを希望しています。

Relaxation scheme of Besse

時間の刻み幅を ∆t とし

tn = n∆t, un = u(tn) (n = 0, 1, 2, ...)

とおく. まず, 一般的な時間発展方程式

∂tu(t) = F (u(t)) (1)

について考える. (1) の差分近似として

un+1 − un

∆t
= F (un)

を考えると, これは陽的に

un+1 = un + ∆tF (un)

と解くことができる.

次に, t = tn+1/2 における (1) の近似として

un+1 − un

∆t
= F (un+1/2), un+1/2 =

un+1 + un

2
(2)

を考えよう (Crank-Nicolson scheme). ここで

u(t ± h) = u(t) ± hu′(t) +
h2

2
u′′(t) + O(h3)

だから

u(t + h) − u(t)

h
= u′(t) + O(h),

u(t + h) − u(t − h)

2h
= u′(t) + O(h2)

に注意する. (2) を変形すると

un+1 = 2un+1/2 − un, (3)

2un+1/2 − ∆tF (un+1/2) = 2un (4)

となる. F が非線形作用素の場合, un から un+1/2 を求めるためには, 非線形方程
式 (4) を解かなければならない.

また, F が線形作用素 F (u) = Au のとき, (4) は

(2I − ∆tA)un+1/2 = 2un

となり, この線形方程式を解くことにより, un から un+1/2 が求まり, さらに (3) に
より un+1 が求まる.

次に, A を線形作用素, g を非線形作用素とし

∂tu = Au + g(u)u (5)

について考えよう. Crank-Nicolson scheme で近似した場合は非線形方程式を解か
なければならないが, それを避けるための工夫をする. まず

un+1 − un

∆t
= Aun+1/2 + g(un)un+1/2, un+1/2 =

un+1 + un

2
(6)

と近似すると, un から, 線形方程式

(2I − ∆tA − ∆tg(un))un+1/2 = 2un

を解くことにより, un+1/2 が求まる. (6)において, g(un)は t = tn における g(u(t))

の近似で, その他の項は t = tn+1/2 における近似になっている. g(u) も t = tn+1/2

において近似したいが, そうすると Crank-Nicolson scheme となってしまう.

そこで, さらに次のような工夫をする (Relaxation scheme of Besse). 新たな未
知関数 ϕ を導入して, 方程式 (5) を{

ϕ = g(u) (7)

∂tu = Au + ϕu (8)

と書き換える. (7) は t = tn で, (8) は t = tn+1/2 で近似すると

ϕn+1/2 + ϕn−1/2

2
= g(un),

un+1 − un

∆t
= Aun+1/2 + ϕn+1/2un+1/2

となる. また, ϕ−1/2 = g(u0) とする. このとき, (ϕn−1/2, un) から

ϕn+1/2 = 2g(un) − ϕn−1/2,

(2I − ∆tA − ∆tϕn+1/2)un+1/2 = 2un

を解くことにより, ϕn+1/2 と un+1/2 が求まる. さらに

un+1 = 2un+1/2 − un

により un+1 が求まり, (ϕn+1/2, un+1) が求まる.

周期境界条件の下での非線形 Schrödinger 方程式の初期値問題
i∂tu = −∂2

xu + V (x)u − g(u)u, 0 ≤ x < 2L, 0 < t ≤ T,

u(t, 0) = u(t, 2L), 0 < t ≤ T,

u(0, x) = u0(x), 0 ≤ x < 2L

に relaxation scheme of Besse を応用してみよう. ここで, V (x) は実数値関数で,

非線形項の典型例は g(u) = c|u|2p (c ∈ R, p > 0) である.

• 空間の区間 [0, 2L) を N 等分する. 空間の刻み幅は ∆x = 2L/N .

• 離散空間変数を xj = j∆x (j = 0, 1, ..., N − 1) とする.

• 時間の刻み幅 ∆t を与え, 離散時間変数を tn = n∆t (n = 0, 1, 2, ...) とする.

• 未知関数 u(t, x) に対して

un = [un
0 , u

n
1 , ..., u

n
N−1]

T , un
j = u(tn, xj)

とおく. ここで, T はベクトルの転置を表す. 各 n に対して, un は N 次元縦
ベクトルである.

• ベクトル u = [u0, u1, ..., uN−1]
T に対して,

g(u) = [g(u0), g(u1), ..., g(uN−1)]
T .

Laplacian の近似

u(x ± h) = u(x) ± hu′(x) +
h2

2
u′′(x) ± h3

3!
u(3)(x) + O(h4)

より
u(x + h) + u(x − h) − 2u(x)

h2
= u′′(x) + O(h2).

そこで, 周期境界条件の下での ∂2
x を N 次実対称行列

Dxx =
1

(∆x)2



−2 1 0 0 0 · · · 1

1 −2 1 0 0 · · · 0

0 1 −2 1 0 · · · 0
...

. . . . . . . . .
...

0 0 · · · 1 −2 1 0

0 0 · · · 0 1 −2 1

1 0 · · · 0 0 1 −2


で近似する. また, A = Dxx − diag [V (x0), V (x1), ..., V (xN−1)] とおく.

方程式の近似 方程式を

wn+1/2 + wn−1/2

2
= g(un), w−1/2 = g(u0)

i
un+1 − un

∆t
+ Aun+1/2 + wn+1/2uu+1/2 = 0, un+1/2 =

un+1 + un

2

と近似する. ここで, un+1 − un = 2(un+1/2 − un) に注意し,

W n+1/2 = diag (wn+1/2)

とおくと
wn+1/2 = 2g(un) − wn−1/2,(

2iIN + ∆tA + ∆tW n+1/2
)
un+1/2 = 2iun

となる. ここで, IN は N 次単位行列である. この連立一次方程式を解けば, un か
ら un+1/2 が求まる. さらに

un+1 = 2un+1/2 − un

により, un+1 が求まる.

数値計算ソフトウェア Scilab を用いたプログラムを 2ページ後に載せた.

保存則 N 次元ベクトル u, v に対して, 内積とノルムを

⟨u, v⟩ = ∆x

N−1∑
j=0

ujvj, ∥u∥ = ⟨u, u⟩1/2

と定める. 上で求めた {un} に対して, L2 ノルムは保存される. すなわち, すべて
の n = 0, 1, 2, ... に対して, ∥un∥2 = ∥u0∥2 が成り立つことが簡単に分る. また, あ
る状況では, エネルギーも保存されることが, [2] で示されている. 正確な定式化及
び証明は [2] を参照していただきたい.

参考文献
[1] C. Besse, Schéma de relaxation pour l’équation de Schrödinger non linéaire et

les systèmes de Davey et Stewartson, C. R. Acad. Sci. Paris Sér. I Math. 326

(1998) 1427–1432.

[2] C. Besse, A relaxation scheme for the nonlinear Schrödinger equation, SIAM

J. Numer. Anal. 42 (2004) 934–952.

Solitary waves for nonlinear Schrödinger equations

非線形 Schrödinger 方程式

i∂tu + ∂2
xu + g(u)u = 0, (x, t) ∈ R2

の孤立波解についてまとめておこう.

• g(u) = c3|u|2p (p > 0, c3 > 0) のとき

ω > 0, k ∈ R をパラメータとする孤立波解

uω,k(t, x) = exp(ikx − ik2t) exp(iω2t)φω(x − 2kt),

φω(x) =

(
ω

d3 cosh(pωx)

)1/p

, d3 =

√
c3

p + 1

が存在する. このとき

uω,k(0, x) = (ω/d3)
1/p exp(ikx){cosh(pωx)}−1/p.

• g(u) = c2|u|p + c3|u|2p (p > 0, c2, c3 ∈ R) のとき

ω > 0, k ∈ R をパラメータとする孤立波解

uω,k(t, x) = exp(ikx − ik2t) exp(iω2t)φω(x − 2kt),

φω(x) =

(
ω2

d2 +
√

d2
2 + d2

3ω
2 cosh(pωx)

)1/p

, d2 =
c2

p + 2
, d3 =

√
c3

p + 1

が存在する. このとき

uω,k(0, x) = exp(ikx)

(
ω2

d2 +
√

d2
2 + d2

3ω
2 cosh(pωx)

)1/p

.

clear; // Nonlinear Schrodinger equation
// iu_t+u_{xx}+c3|u|^{2p}u=0, 0<x<2L, 0<t<T
// with periodic boundary conditions

// Constants
c3=20; p=1; 
L=10; N=256; 
T=10; NT=20; dt=10^(-3); 
ome1=4; k1=0; ome2=3; k2=-1; 

dx=2*L/N; x=[0:N-1]'*dx; // Space variable

Dxx=-2*speye(N,N)+[spzeros(1,N);speye(N-1,N)]...
+[spzeros(N,1),speye(N,N-1)];  
Dxx(1,N)=1; Dxx(N,1)=1;
Dxx=Dxx/dx/dx; // Second Derivative in x 

function y=g(u) // Nonlinear term
  y=c3*abs(u)^(2*p)
endfunction

d3=sqrt(c3/(p+1)); // Solitary waves 
function y=f1(x)
  y=(ome1/d3)^(1/p)*exp(%i*k1*x).*(cosh(p*ome1*x))^(-1/p)
endfunction
function y=f2(x)
  y=(ome2/d3)^(1/p)*exp(%i*k2*x).*(cosh(p*ome2*x))^(-1/p)
endfunction

u0=f1(x-L+4)+f2(x-L-4); // Initial data

IA=2*%i*speye(N,N)+dt*Dxx;
timer(); // Relaxation scheme of Besse 
t=0; u=u0; w=g(u);
for n=1:NT,
  while t<=n*T/NT, 
    w=2*g(u)-w; 
    v=(IA+dt*sparse(diag(w)))\(2*%i*u);
    u=2*v-u; clear v; 
    t=t+dt;
  end 
  clf(); // Animation 
  plot(x,real(u),'g'); plot(x,imag(u),'y'); 
  plot(x,abs(u),'b'); plot(x,-abs(u),'b');
  Mass(n)=sum(abs(u)^2)*dx; 
  Energy(n)=-real(u'*Dxx*u)*dx/2-c3*sum(abs(u)^(2*p+2))*dx/(2*p+2);
end
Timer=timer()

plot(x,abs(u0),'r');
xset("window",1)
plot(Mass);
xset("window",2)
plot(Energy);

ErrorMass=max(Mass)-min(Mass)
ErrorEnergy=max(Energy)-min(Energy)

Page : 1



Splitting method for nonlinear Schrödinger equations

まず, splitting method の一般的なアイデアは, 発展方程式

∂tu = A(u) + B(u), u(0) = u0

の解 u(t) = S(t)u0 を, ∂tv = A(v) の解作用素 X(t) と ∂tw = B(w) の解作用素
Y (t) の組み合わせで近似することである.

ZL(t) = Y (t)X(t), ZS(t) = Y (t/2)X(t)Y (t/2)

とおく. このとき, T > 0 に対して, ∆t > 0 を十分小さくとれば, n∆t ≤ T なる
n ∈ N に対して

∥S(n∆t)u0 − ZL(∆t)nu0∥ ≤ C∆t,

∥S(n∆t)u0 − ZS(∆t)nu0∥ ≤ C(∆t)2

が成り立つことを期待する. ここで, C は T と u0 に依存する正定数である. ZL

は Lie formula, ZS は Strang formula と呼ばれる. また, Y (t) が可逆であるとき

ZS(∆t)n = Y (−∆t/2)ZL(∆t)nY (∆t/2)

に注意する.

このアイデアを, 非線形 Schrödinger 方程式の初期値問題

(1)

{
i∂tu = −∂2

xu + V (x)u − g(u)u (t, x) ∈ R2

u(0, x) = u0(x) x ∈ R

に適用してみよう. ここで, V (x) は実数値関数で, 非線形項 g は, g(z) = g(|z|)
(z ∈ C) をみたす実数値関数とする. 典型例は g(u) = c|u|2p (c ∈ R, p > 0) である.

まず, (1) を

(2)

{
∂tv = i∂2

xv (t, x) ∈ R2

v(0, x) = v0(x) x ∈ R

(3)

{
∂tw = i[g(w) − V (x)]w (t, x) ∈ R2

w(0, x) = w0(x) x ∈ R

に分解する. (2) の解は

v(t, x) = F−1[e−itξ2Fv0](x)

で与えられる. また, (3) の解は

w(t, x) = exp
(
it{g(w0(x)) − V (x)}

)
w0(x) (4)

で与えられる. 実際, g は実数値関数だから, 任意の x ∈ R に対して

∂t|w(t, x)|2 = 2ℜ(ww) = 2ℜ{i[g(w(t, x)) − V (x)]|w(t, x)|2} = 0.

よって, すべての (t, x) ∈ R2 に対して |w(t, x)| = |w0(x)|. また, g(z) = g(|z|) だか
ら, すべての (t, x) ∈ R2 に対して g(w(t, x)) = g(w0(x)) となり, (4) が得られる.

数値計算ソフトウェア Scilab を用いたプログラムを最後のページに載せた.

26–40.

[2] H. Holden, K. H. Karlsen, K.-A. Lie, and N. H. Risebro, Splitting methods

for partial differential equations with rough solutions, Analysis and MATLAB

programs, EMS Series of Lectures in Mathematics, European Mathematical

Society (EMS), Zürich, 2010.

離散 Fourier 変換

f(x) は周期 2L の周期関数とする. このとき, k ∈ Z に対し, ξk = kπ/L とおき,

f(x) の Fourier 係数を

f̂(ξk) =
1

2L

∫ 2L

0

f(x)e−iξkx dx (1)

と定めると

f(x) =
∞∑

k=−∞

f̂(ξk)e
iξkx (2)

が成り立つ. これを離散化しよう.

• N を 2のべき乗とし,基本区間 [0, 2L)を N 等分する. 刻み幅は∆x = 2L/N .

• 離散空間変数を xj = j∆x とし, fj = f(xj) とおく (j ∈ Z).

このとき, fj+N = fj (j ∈ Z)だから,関数 f(x)は N 次元ベクトル [f0, f1, ..., fN−1]

に離散化される. (1) の積分を台形公式で近似すると

f̂(ξk) ∼
1

2L

N−1∑
j=0

f(xj)e
−iξkxj∆x = f̂k

となる. ここで

f̂k =
1

N

N−1∑
j=0

fjω
−jk, ω = exp

(
2πi

N

)
とおいた. このとき, ωN = 1 より, f̂k+N = f̂k (k ∈ Z) だから, f̂(ξk) は N 次元ベ
クトル [f̂0, f̂1, ..., f̂N−1] または [f̂−N/2+1, ..., f̂−1, f̂0, f̂1, ..., f̂N/2] に対応する.

また, (2) の右辺の無限級数を

N/2∑
k=−N/2+1

f̂kω
jk =

N−1∑
k=0

f̂kω
jk

で近似する. このとき, すべての j = 0, 1, ..., N − 1 に対して

N−1∑
k=0

f̂kω
jk =

1

N

N−1∑
k=0

N−1∑
l=0

flω
(j−l)k =

1

N

N−1∑
l=0

fl

N−1∑
k=0

ω(j−l)k

で, j, l = 0, 1, ..., N − 1 に対して, |j− l| ≤ N − 1 だから, l ̸= j のとき, ωj−l ̸= 1 で

N−1∑
k=0

ω(j−l)k =
1 − ω(j−l)N

1 − ωj−l
= 0.

また, l = j のとき,
N−1∑
k=0

ω(j−l)k = N だから, (2) の対応物として

fj =
N−1∑
k=0

f̂kω
jk =

N/2∑
k=−N/2+1

f̂kω
jk (j ∈ Z)

が得られた. さらに, (2) を微分すると

f ′(x) =
∞∑

k=−∞

iξkf̂(ξk)e
iξkx

だから, f ′(x) は

N/2∑
k=−N/2+1

iξkf̂kω
jk =

N/2∑
k=0

iξkf̂kω
jk +

N−1∑
k=N/2+1

iξk−N f̂kω
jk

で近似される.

Scilab では, N 次元ベクトル u = [u0, u1, ..., uN−1] に対して, v=fft(u) は N 次
元ベクトル

v = [v0, v1, ..., vN−1], vk =
N−1∑
j=0

ujω
−jk

を表す. また, N 次元ベクトル v = [v0, v1, ..., vN−1] に対して, u=ifft(v) は N 次
元ベクトル

u = [u0, u1, ..., uN−1], uj =
1

N

N−1∑
k=0

vkω
jk

を表す. fft と ifft は互いに逆変換である. また

xi = [ξ0, ξ1, ..., ξN/2, ξ−N/2+1, ..., ξ−1] =
π

L
[0, 1, ..., N/2,−N/2 + 1, ...,−1]

とおけば, 微分演算 p(∂x)u(x) は ifft(p(%i*xi).*fft(u)) で近似される. ここ
で, u = [u(x0), u(x1), ..., u(xN−1)], %i は虚数単位で, 関数 p(z) とベクトル v =

[v0, v1, ..., vN−1] に対して, p(v) はベクトル

p(v) = [p(v0), p(v1), ..., p(vN−1)]

を表し, ベクトル v と w に対して, v. ∗ w はベクトル

v. ∗ w = [v0w0, v1w1, ..., vN−1wN−1]

を表す.

clear; // Nonlinear Schrodinger equation 
// iu_t+u_{xx}+c_3|u|^{2p}u=0, 0<x<2L, 0<t<T
// with periodic boundary conditions

// Constants 
c3=20; p=1; 
L=10; N=256; 
T=10; NT=20; dt=10^(-4); 
ome1=4; k1=0; ome2=3; k2=-1; 

dx=2*L/N; x=[0:N-1]*dx; // Space variable
xi=[0:N/2,-N/2+1:-1]*%pi/L; // Fourier variable

function y=g(u) // Nonlinear term
  y=c3*abs(u)^(2*p)
endfunction

function v=Dx(u) // Derivative in x
  v=ifft(%i*xi.*fft(u))
endfunction

d3=sqrt(c3/(p+1)); // Solitary waves 
function y=f1(x)
  y=(ome1/d3)^(1/p)*exp(%i*k1*x).*(cosh(p*ome1*x))^(-1/p)
endfunction
function y=f2(x)
  y=(ome2/d3)^(1/p)*exp(%i*k2*x).*(cosh(p*ome2*x))^(-1/p)
endfunction

u0=f1(x-L+4)+f2(x-L-4); // Initial data

timer(); // Splitting method with Strang formula 
t=0; u=u0; 
for n=1:NT,
  u=exp(%i*dt/2*g(u)).*u;
  while t<=n*T/NT,
    u=ifft(exp(-%i*dt*xi^2).*fft(u));
    u=exp(%i*dt*g(u)).*u;
    t=t+dt;
  end 
  u=exp(-%i*dt/2*g(u)).*u;  
  clf(); // Animation 
  plot(x,real(u),'g'); plot(x,imag(u),'y');
  plot(x,abs(u),'b'); plot(x,-abs(u),'b');
  Mass(n)=sum(abs(u)^2)*dx; 
  Energy(n)=sum(abs(Dx(u))^2)*dx/2-c3*sum(abs(u)^(2*p+2))*dx/(2*p+2);
  end
Timer=timer()

plot(x,abs(u0),'r');
xset("window",1)
plot(Mass);
xset("window",2)
plot(Energy);

ErrorMass=max(Mass)-min(Mass)
ErrorEnergy=max(Energy)-min(Energy)

